

Multivariable Limits

©2006 by Mike May, S.J.- maymk@slu.edu

This worksheet is intended to look at the material on limits for functions of 2 variables.

[> restart;

▼ Functions of one variable, a review:

▼ The definition in one variable

Before looking at the definitions of limit and continuity for functions of several variables, it is worthwhile to review them for functions of one variable.

When we say that the limit as x approaches a of $f(x)$ is L we mean that...

for every $\epsilon > 0$ there is a $\delta > 0$ such that $|f(x) - L| < \epsilon$ whenever $0 < |x - a| < \delta$.

That is quite a mouthful. To put it into more visual terms, we are claiming that the point (a, L) belongs to the natural continuation of the graph of $y=f(x)$. We test the claim by putting a box around the point (a, L) , going up and down by ϵ and right and left by δ . We claim that no matter the ϵ chosen we can choose δ so that the graph exits through the sides of the viewing box rather than through the top and bottom.


[>

▼ Example 1, Demonstrating a limit:

We claim that as x approaches 2, the limit of $x^2 + 3x$ is 10. To prove this claim we would need a rule for finding a δ for every ϵ greater than 0. We will be satisfied with finding a δ when ϵ is .01.


We start by trying δ at .01 as well.

```
> f := x -> x^2+3*x;
a := 2; L:= 10:
eps := .01:
del := .01:
plot(f(x), x=a-del..a+del, y=L-eps..L+eps, axes=boxed);
      f:=x→x2+3x
      a:=2
```



It is clear from the graph that our δ is too big. Next we try a δ of .001.

```
> f := x -> x^2+3*x;
a := 2; L:= 10:
eps := .01:
del := .001:
plot(f(x), x=a-del..a+del, y=L-eps..L+eps, axes=boxed);
   $f := x \rightarrow x^2 + 3x$ 
   $a := 2$ 
```



That value of δ works. Since we have zoomed in enough to make the graph look like a line we suspect that setting δ to $\frac{\epsilon}{10}$ will work for smaller ϵ s. That is a problem for another day however.

[>


Example 2, Disproving a limit:

The reverse problem is to show that something is not a limit. To do that we need to find an ϵ for which no δ is small enough. In the example above, suppose someone claimed that the limit is 11 rather than 10. We want to find a y range around 11 that the graph always escapes no matter how small the x -range around 2 is. We will look at $\epsilon = 0.5$ and $\delta = 2$.

```
> f := x -> x^2+3*x;
a := 2; L:= 11:
eps := .5:
del := .2:
plot(f(x), x=a-del..a+del, y=L-eps..L+eps, axes=boxed);
```

$$f:=x \rightarrow x^2 + 3x$$

$$a := 2$$



It is clear that making the box narrower will not clean up the problems at 2.

[>

Example 3, Showing a function has no limit:


A harder problem is to show there is no limit at a point. This usually means that the graph is either going off to infinity or that it is trying to get close to two different points. In that case, we make ϵ equal to $1/3$ the distance between the two y values so that no box includes them both.

Consider the function $f(x) = \text{abs}(x)/x$. Looking at the graph we see that as x gets close to 0 we need to include both 1 and -1. If ϵ is $2/3$, there is no L we can choose to put both 1 and -1 in the range $[L-2/3, L+2/3]$.

```
> f := x -> abs(x)/x;
a := 0; L:= 0:
eps := .67:
del := .1:
plot(f(x), x=a-del..a+del, y=L-eps..L+eps, axes=boxed,
discont=true);
```

$$f := x \rightarrow \frac{|x|}{x}$$

$$a := 0$$



[>

▼ Exercises:

1) Give evidence that the limit of $\frac{x^3 - 8}{x - 2}$ as x approaches 2 is 12 by finding values of delta that work in the definition when epsilon is .1, .01, and .001.

[>

2) Explain why $\sin\left(\frac{1}{x}\right)$ has no limit as x approaches 0. (Hint, you want to explain why a box around the limit need to include the y -values 1 and -1.)

[>

[>

▼ Limits of functions in 2 variables:

▼ The easy generalization of the definition

For functions of two variables we can proceed by making minor modifications in the previous definition. It becomes:

When we say that the limit as (x, y) approaches (a, b) of $f(x, y)$ is L we mean that...

for every $\epsilon > 0$ there is a $\delta > 0$ such that $|f(x, y) - L| < \epsilon$ whenever $(x, y) \neq (a, b)$, $|x - a| < \delta$, and $|y - b| < \delta$.

That is even more of a mouthful. To put it into more visual terms, we are claiming that the point (a, b, L) belongs to the natural continuation of the graph of $y=f(x, y)$. We test the claim by putting a box around the point (a, b, L) , going up and down by ϵ and right, left, forward, and back by δ . We claim that no matter the ϵ chosen we can choose δ so that the graph exits through the sides of the viewing box rather than through the top and bottom.

(Advanced note: The definition is slightly different from what is given in most textbooks. We are using square neighborhoods rather than round neighborhoods because square neighborhoods are easier to draw.)


>

Example 1, an easy limit:

The normal case we will deal with is a function that has a hole we need to fill in. We want to look at the graph and see that we have an obvious limit.

```
> f := (x,y) -> sin(x^2+y^2)/(x^2+y^2);  
a := 0: b:= 0: L:= 1:  
eps := .1:  
del := 0.5:  
plot3d(f(x,y), x=a-del..a+del, y=b-del..b+del,  
view=L-eps..L+eps, axes=boxed, style=patchcontour);
```

$$f := (x, y) \rightarrow \frac{\sin(x^2 + y^2)}{x^2 + y^2}$$



The graph appears to be flat, leading us to guess that choosing δ equal to ϵ will work in the definition. You can verify that this will work if ϵ is .1, .01, or .001. In fact this function is so well behaved that you should try a δ of 1 with an ϵ of .2 to see that not every delta works.

As in the single variable case, this does not prove the function has 1 as a limit when the input approaches the origin, but it does leave us pretty convinced of the fact.

>


Example 2, an easy nonlimit:

Now we turn to a function that does not have a limit at the origin.

Note that we look at a bigger patch when trying to understand a more confusing function.

```
> f := (x,y) -> (x*y)/(x^2+y^2);
a := 0: b:= 0: L:= 0:
eps := 1.2:
del := 1:
plot3d(f(x,y), x=a-del..a+del, y=b-del..b+del, view=L-eps..
L+eps, axes=boxed,style=patchcontour);
```


$$f := (x, y) \rightarrow \frac{xy}{x^2 + y^2}$$



The picture makes us suspicious that z-values ranging from -0.5 to 0.5 occur arbitrarily close to the origin. That means that when ϵ is 0.4 , our box is only 0.8 high, but it needs to cover a y -range that is 1 high. No δ will be small enough.

```
> f := (x,y) -> (x*y)/(x^2+y^2);
a := 0: b:= 0: L:= 0:
eps := 0.4:
del := .1:
plot3d(f(x,y), x=a-del..a+del, y=b-del..b+del, view=L-eps..
L+eps, axes=boxed,style=patchcontour);
```

$$f := (x,y) \rightarrow \frac{xy}{x^2 + y^2}$$



You should verify that the picture does not get any better if we make delta .01 or .001. It looks like there is no "right way" to fill in this function at the origin. The function does not have a limit at the origin.


[>

Example 3, a hard nonlimit- testing paths:

In our last example we had lines coming into our problem point (the origin) that have different limits. Sometimes when we look at the graph we see that the contours going into the origin are not lines. Consider the following function.

```
> f := (x,y) -> (x^2*y)/(x^4+y^2);
a := 0: b:= 0: L:= 0:
eps := 1.2:
del := 1:
plot3d(f(x,y), x=a-del..a+del, y=b-del..b+del, view=L-eps..
L+eps, axes=boxed,style=patchcontour);
```


$$f := (x, y) \rightarrow \frac{x^2 y}{x^4 + y^2}$$



The contours get confused near the origin. (The contours look like they are trying to be parabolas until they get too close to the x-axis.) That means that we need to be careful about simple zooming in. Our grapher may miss interesting features. Consider the following zoom.

```
> f := (x,y) -> (x^2*y)/(x^4+y^2);
a := 0: b:= 0: L:= 0:
eps := 1.2:
del := .01:
plot3d(f(x,y), x=a-del..a+del, y=b-del..b+del, view=L-eps..
L+eps, axes=boxed,style=patchcontour);
```

$$f := (x,y) \rightarrow \frac{x^2 y}{x^4 + y^2}$$




When we zoomed in all the contours disappear.

When we see strange things happening in a graph we should be suspicious that simply zooming in may miss the interesting features. One of the things we can do is look at what happens along a specified path. (Time to remember parametric curves from precalculus.) The contours of the original graph look like they are trying to be parabolas. Thus we will look at the graphs of parabolas, that is, we evaluate the function along parabolic paths.

```
> f := (x,y) -> (x^2*y)/(x^4+y^2);
a := 0: b:= 0: L:= 0:
eps := 1.2:
del := .01:
plot({f(x,x^2),f(x,2*x^2),f(x,-.2*x^2)}, x=a-del..a+del,
view=L-eps..L+eps, axes=boxed);
```

$$f := (x, y) \rightarrow \frac{x^2 y}{x^4 + y^2}$$



This makes it clear that the contours are parabolas. Since the origin would need to be on all these level curves, there is no right value to give at the origin and the function has no limit there.

[>

Exercises:

3) Let $f(x, y) = x*y*(x^2-y^2)/(x^2+y^2)$ when $(x,y) \neq (0,0)$. Justify the claim that the limit of $f(x, y)$ as (x,y) approaches $(0,0)$ is 0 by finding δ s that work in the definition of limit when ϵ is .1, .01, and .001.

[>

4) Let $f(x, y) = \exp(-x-y)$. Justify the claim that the limit of $f(x,y)$ as (x,y) approaches $(0,0)$ is 1 by finding δ s that work in the definition of limit when ϵ is .1, .01, and .001.

[>

5) Let $f(x, y) = (x^2-y^2)/(x^2+y^2)$ when $(x,y) \neq (0,0)$. Show that the function does not have a limit at the origin.

[>

6) Let $f(x, y) = (x+y)/(x-y)$ when $x \neq y$. Show that the function does not have a limit at the origin.

[>

[>